

Problem Set 3

The Game of Twenty-One

Issued: Thursday, 5 October 2000

Due: Tuesday, 10 October 2000

Reading: Text (SICP 2nd Edition by Abelson & Sussman): Sections 2.1 and 2.2

Louis Reasoner took a course on game theory and became interested in the card game Twenty-One (also called Blackjack). Louis was also treasurer of his living group. By the end of the semester, he had managed to squander the term's dinner money at Atlantic City casinos in an attempt to perfect his "no-lose" strategy. Ben Bitdiddle, Louis's roommate, has decided to construct a general-purpose Twenty-One simulator to help discover what Louis has been doing wrong (in terms of playing Twenty-One, not in terms of moral or ethical conduct).

For our purposes, the rules of Twenty-One are as follows. There are two players, and the object of the game is to be dealt a set of cards that totals as close to 21 as possible without going over 21, where each number card has that number as its value, and face cards have a value of 10. Each player is dealt one card face up that the other player can see. These are called the up cards. Subsequent cards are dealt face down. One player plays first, asking for more cards one at a time (called a *hit*) until he decides to *stay* with the total he has or until his total exceeds 21. If a player's total exceeds 21 he *busts*, meaning he immediately loses the game. If the first player does not bust, the second player, called the *house*, then plays, asking for more cards until either losing by exceeding 21 or deciding to stay with the current total. After the house decides not to take additional cards both players expose their cards, and the player with the largest total wins. In the event of a tie, the house wins. This version of Twenty-One is simplified: we will not consider such things as "splitting" or the special treatment of aces. In fact, since we are interested only in the relative strengths of competing strategies, we will not simulate betting either.

A player's *strategy* determines when he wishes another card and when he would like to stay with what he has. For our Scheme simulation, each player will be modeled by a procedure that implements his strategy. Since a typical strategy for when to stay and when to hit involves both the player's current hand and the point value of the opponent's face-up card, we will represent a strategy as a procedure of two arguments: the player's hand and the point value of the opponent's face-up card. The procedure returns true if the player would want another card, and false if the player would stay. For example, the following (stupid) strategy procedure will always take a card if the opponent's up card is greater than 5:

```
(define (stupid-strategy my-hand opponent-up-card)
  (> opponent-up-card 5))
```

The following procedure `play-hand` takes as arguments a strategy, a hand, and the opponent's up card. It continues to accept cards for as long as the strategy requests, or until the total of the cards in the hand exceeds 21. `Play-hand` returns as a value, the full hand that was dealt.

```
(define (play-hand strategy my-hand opponent-up-card)
  (cond ((> (hand-total my-hand) 21) my-hand) ; I lose... give up
        ((strategy my-hand opponent-up-card) ; hit?
         (play-hand strategy
                     (hand-add-card my-hand (deal))
                     opponent-up-card))
        (else my-hand))) ; stay
```

For the purposes of this simple simulation, a “hand” of cards will be represented by two numbers—the value of the up card and the total of all the cards in the hand. We will represent this using a very simple form of *data abstraction* – we have a *constructor* procedure `make-hand` that creates a hand from two numbers, and two *selectors* `hand-up-card` and `hand-total` that return the up card and total of a given hand:

```
(define (make-hand up-card total)
  (cons up-card total))

(define (hand-up-card hand)
  (car hand))

(define (hand-total hand)
  (cdr hand))
```

Exercise 1 Draw the box and pointer diagram resulting from the evaluation of

```
(make-hand 10 15)
```



In terms of these basic procedures, we can implement some useful operations on hands. `Make-new-hand` takes as argument a first card and returns a hand containing only that card (i.e., the card is both the up card and the total):

```
(define (make-new-hand first-card)
  (make-hand first-card first-card))
```

`Hand-add-card` takes a hand and a new card and returns a hand with the same up-card as the original, but with the total augmented by the value of the new card:

```
(define (hand-add-card hand new-card)
  (make-hand (hand-up-card hand)
             (+ new-card (hand-total hand))))
```

Instead of modeling a real deck of cards, we simply deal cards at random from an infinite deck in which each card value from 1 to 10 is equally probable. (This does not, of course, correctly model real decks of cards, but since our focus is one strategies, we won’t worry about the difference.) We represent dealing a card as simply returning a random number in the range 1 through 10:

```
(define (deal) (+ 1 (random 10)))
```

Finally, the top-level procedure in our simulation, `twenty-one`, simulates one game of Twenty-One. It takes strategy procedures for a player and for the house as its two arguments. It creates initial hands for the house and the player, then plays the player strategy, then plays the house strategy. `Twenty-one` returns 1 if the player wins the simulated game and 0 if the house wins.

```

(define (twenty-one player-strategy house-strategy)
  (let ((house-initial-hand (make-new-hand (deal)))) ; set up house hand
    ; let is covered on pp.64-66 of text
    (let ((player-hand ; set up initial hand, and play out
          (play-hand player-strategy ; strategy to use
                     (make-new-hand (deal)) ; initial player hand
                     (hand-up-card house-initial-hand)))) ;
      ;information about house hand available to player
      (if (> (hand-total player-hand) 21)
          0 ; 'bust': player loses
          (let ((house-hand ; play out house hand
                (play-hand house-strategy
                           house-initial-hand
                           (hand-up-card player-hand))))
            (cond ((> (hand-total house-hand) 21)
                  1) ; 'bust': house loses
                  ((> (hand-total player-hand)
                      (hand-total house-hand))
                   1) ; house loses
                  (else 0)))))) ; player loses

```

Hit? is a simple interactive strategy procedure that can be used with `twenty-one`. It displays on the screen the information available to the player it is simulating and asks whether it should take another card. It returns true if you type y and false if you type any other character.

```

(define (hit? your-hand opponent-up-card)
  (newline)
  (display "Opponent up card ")
  (display opponent-up-card)
  (newline)
  (display "Your Total: ")
  (display (hand-total your-hand))
  (newline)
  (display "Hit? ")
  (user-says-y?))

```

Exercise 2 Load in the code for Problem Set 3 and try playing a few games of Twenty-One against yourself by evaluating:

```
(twenty-one hit? hit?)
```



Remember that the first set of questions you will be asked are for the player's hand and the second set of questions are for the house's hand. There is nothing to turn in for this problem.

Exercise 3 Define a procedure `stop-at` that takes a number as argument and returns a strategy procedure. The strategy `stop-at` should ask for a new card if and only if the total of a hand less than the argument to `stop-at`. For example `(stop-at 16)` should return a strategy that asks for another card if the hand total is less than 16, but stops as soon as the total reaches 16. To test your implementation of `stop-at`, play a few games by evaluating

```
(twenty-one hit? (stop-at 16))
```



Thus, you will be playing against a house whose strategy is to stop at 16. Turn in a listing of your procedure.

Exercise 4 Define a procedure `test-strategy` that tests two strategies by playing a specified number of simulated Twenty-One games using the two strategies. `Test-strategy` should return the number of games that were won by the player (and thus lost by the house). For example,

```
(test-strategy (stop-at 16) (stop-at 15) 100)
```

should play one hundred games of Twenty-One, using the value returned by `(stop-at 16)` as the player's strategy and the value of `(stop-at 15)` as the house strategy. It should return a non-negative integer indicating how many games were won by the player. Turn in a listing of your procedure and some sample results.



Exercise 5 When the simulated games in the previous exercise ran, it was impossible for us to tell what was going on. It would be nice if we could watch a strategy play by observing its inputs and the decisions it makes. Define a procedure called `watch-player` that takes a strategy as an argument and returns a strategy as its result. The strategy returned by `watch-player` should implement the same result as the strategy that was passed to it as an argument, but, in addition, it should print the information supplied to the strategy and the decision that the strategy returns. For example,

```
(test-strategy (watch-player (stop-at 16))
               (watch-player (stop-at 15))
               2)
```



should play two simulated games and show what each player does at each step. Turn in a listing of your procedure and some sample runs using it. You will need to use the `display` procedure.

Exercise 6 Ben has finally gotten Louis to describe his Twenty-One strategy. Here is how Louis was playing. If his hand contained fewer than 12 points, he always asked for another card. If his hand had more than 16 points, he always stayed with what he had. If his hand had exactly 12 points, he took another card if his opponent's up card was less than 4. If his hand had exactly 16 points, he would stay if his opponent was showing a 10. If none of the above conditions held, Louis would take a card if his opponent's up card was greater than 6, otherwise he would stay with what he had. Define a procedure called `louis` that implements Louis's strategy. Try Louis's strategy against the strategies of stopping at 15, 16, and 17 by evaluating

```
(test-strategy louis (stop-at 15) 100)
(test-strategy louis (stop-at 16) 100)
(test-strategy louis (stop-at 17) 100)
```



Exercise 7 Implement a procedure `both` that takes two strategies as arguments and returns a new strategy. This new strategy will call for a new card if and only if *both* strategies would ask for a new card. For example, using the strategy

```
(both (stop-at 19) hit?)
```



will ask for a new card only if the hand total is less than 19 and the user requests a hit. Turn in a listing of your procedure and an example showing that it works.

Exercise 8 The simulation above is very restricted in that it represents a hand simply as a pair of numbers: the up card and the total value of the hand. Suppose we also want to keep track of all of the cards that a player has received.

We could rewrite our constructor `make-new-hand` as follows:

```
(define (make-new-hand first-card)
  (make-hand first-card first-card (list first-card)))
```

Type this into your answer file for the problem set and evaluate it to rebind `make-new-hand` to this new definition.

Note that this calls `make-hand`. We will also need to update that constructor. Complete the following definition and insert it into your answer file.

```
(define (make-hand up-card total card-list)
  ???)
```

Exercise 9 We also need to change the way that we add a card to our hand. Rewrite `hand-add-card`. It should still take `hand` and `new-card` as parameters. (Be sure to keep the order of the parameters the same as in the previous version of the code. This will allow us to change the representation of a hand without having to change our calls to `hand-add-card`.)

Exercise 10 Show the box and pointer diagram that results from the evaluation of

```
(define my-hand (make-new-hand 5))
```



Exercise 11 Show the box and pointer diagram that results from the evaluation of

```
(define my-hand-after-new-card (hand-add-card my-hand 10))
```

assuming that `my-hand` is the hand defined in Exercise 10.



Exercise 12 Write the selectors for our new representations of a hand. You'll need to write `hand-up-card`, `hand-total` and `list-of-cards`.

Exercise 13 Do you need to change `twenty-one` to reflect these new constructors? If no, why? If yes, make the appropriate change(s). Verify that everything works by running `(twenty-one hit? hit?)`.



Exercise 14 Now that we have the extra information stored in a hand, let's display it for the user. Modify `hit?` to print out the list of the cards in the hand after it prints out the user's card total. Show a sample run of your code for `(twenty-one hit? hit?)`.



Prepare for tutorial In our simulation, we used an infinite deck in which each type of card from 1 to 10 is equally probable. Ben Bitdiddle is perturbed by this. He complains, "With an infinite deck of equally probable card values, I can't use my winning card-counting strategy."

(a) How would you implement a procedure that generates a fresh deck of cards that looks like:

```
(1 1 1 1 2 2 2 2 3 3 3 3 ... 10 10 10 10)
```

How would you modify this procedure to generate a deck that also includes face cards (Kings, Queens, and Jacks) given that each face card has a value of 10?

(b) How would you implement a `shuffle` procedure that accepts a deck as an argument and produces a new deck with the same cards as the argument deck but in permuted order? For simplicity, `shuffle` should permute a deck by first cutting the argument deck into two equal-length halves then combining these halves into a new deck by alternately choosing a card from each of the two halves.

(c) Note that the above `shuffle` would always produce the same deck given the same argument. This behavior is called deterministic because it will execute the same way every time. A better `shuffle` would introduce an element of randomness in the shuffle. How would you implement `random-shuffle` that accepts a deck as an argument and cuts it as before but that combines the two halves by alternately choosing some number of cards from each half where the number of cards chosen varies randomly from 1 to 5? (Warning: be careful what you do if you choose more cards than remain in the half-deck.)

(d) Finally, how would you change the simulation to use a finite deck? (Hint: pass around a finite deck as an additional argument to the appropriate procedures and consider implementing procedures for selecting the top card of a deck and for returning the rest of the deck.) Ideally, we could test strategies by starting with a fresh deck, shuffling it a few times, then playing games until we run out of cards. When the deck runs out, we can arbitrarily declare that game a player win.

Problem Set Evaluation To help us evaluate how well the course is going, and how good a job we are doing, please tear off this page, answer the following questions, and attach it to your problem set when you hand it in. While we do not require you to fill in this survey, doing so will help correct any problems and improve the course not only for future students, but for you as well.

Number of hours spent on the reading assignment: _____

Number of total hours spent working directly on the problem set _____

Number of collaborative hours spent working directly on the problem set _____

How hard was this problem set (1–10: 1, piece of cake; 5, just right; 10, far too hard) _____

Hardest problem _____

Fraction of time spent on hardest problem, multiplied by 10 (1–10: 1, only a small fraction; 5, half of the time; 10, every single minute) _____

Coherence between problem set contents and lecture/recitation contents (1–10: 1, seemingly random; 5, good correspondence; 10, perfect coherence) _____

Coherence between problem set contents and assigned reading contents (using the same scale) _____

How much did you enjoy this problem set (1–10: 1, nearly walked out of the program; 5, neutral; 10, recruited other students because it was so good) _____

How much did you enjoy the lectures and recitations for this problem set (using the same scale) _____

You may also submit fully anonymous comments via the electronic means described in class.